

Suppose you are conducting an experiment where you have 2 breeds of ducks and are testing 3 different diets. The statistical model you will use to analyse your data is:

$$Y_{ij} = \mu + \text{breed}_i + \text{diet}_j + \text{breed}_i * \text{diet}_j + e_{ijk}$$

You would like to build a contrast that looks at differences between the two breeds on diet #1 represented by the following Null Hypothesis:

$$H_0: \mu_{D1BR1} = \mu_{D1BR2}$$

Remember that the μ_{D1BR1} is actually made up of three parts:

$$\mu_{D1BR1} = \mu + \mu_{D1} + \mu_{BR1} + \mu_{D1BR1}$$

To build the contrast follow these steps:

1. Write out the contrast

$$\mu + \mu_{D1} + \mu_{BR1} + \mu_{D1BR1} = \mu + \mu_{D1} + \mu_{BR2} + \mu_{D1BR2}$$

2. Bring all terms to one side of the equation:

$$\mu + \mu_{D1} + \mu_{BR1} + \mu_{D1BR1} - \mu - \mu_{D1} - \mu_{BR2} - \mu_{D1BR2} = 0$$

3. Group like terms:

$$\mu - \mu + \mu_{D1} - \mu_{D1} + \mu_{BR1} - \mu_{BR2} + \mu_{D1BR1} - \mu_{D1BR2} = 0$$

4. Reduce the equation:

$$\mu - \mu + \mu_{D1} - \mu_{D1} + \mu_{BR1} - \mu_{BR2} + \mu_{D1BR1} - \mu_{D1BR2} = 0$$

$$\mu_{BR1} - \mu_{BR2} + \mu_{D1BR1} - \mu_{D1BR2} = 0$$

5. Add in all interactions with 0 coefficients

$$\mu_{BR1} - \mu_{BR2} + \mu_{D1BR1} - \mu_{D1BR2} + 0\mu_{D2BR1} - 0\mu_{D2BR2} + 0\mu_{D3BR1} - 0\mu_{D3BR2} = 0$$

6. SAS coding:

Estimate 'Breed 1 vs. Breed 2 in Diet 1' breed 1 -1

```
diet*breed 1 -1 0 0 0 0;
```

contrast 'Breed 1 vs. Breed 2 in Diet 1' breed 1 -1

```
diet*breed 1 -1 0 0 0 0;
```

SAS example:

```
Data test;
  input diet breed wt;
  datalines;
1 1 12
1 1 11
1 1 12
1 2 15
1 2 16
1 2 17
2 1 21
2 1 19
2 1 20
2 2 15
2 2 16
2 2 17
3 1 13
3 1 14
3 1 12
3 2 17
3 2 18
3 2 16
;
Run;

Proc glm;
  class diet breed;
  model wt = diet breed diet*breed;
  estimate 'Breed 1 vs. Breed 2 in Diet 1' breed 1 -1
  diet*breed 1 -1 0 0 0 0;
  contrast 'Breed 1 vs. Breed 2 in Diet 1' breed 1 -1
  diet*breed 1 -1 0 0 0 0;
Run;
Quit;
```